The first turtle graphic discussed in Mindstorms: Children, Computers, and Powerful Ideas by Seymour Papert is a simple drawing of a house. It is a square with a triangle on top for the roof.

For a slightly more advanced audience, a more practical introduction to turtle graphics might be to draw a bar chart.

See image here: https://i.imgur.com/B7YbTbZ.png

Task

 Create a function (or subroutine) that uses turtle graphics to draw a house of a specified size as described above. Optionally make it lovely by adding details such as, for example, doors and windows.
 Create a function (or subroutine) that takes a list (array, vector) of non-negative numbers and draws a bar chart from them, scaled to fit exactly in a square of a specified size. The enclosing square need not be drawn. 
 Both functions should return the turtle to the location it was at and facing in the same direction as it was immediately before the function was executed. 
to rectangle :width :height
    repeat 2 [
        forward :height
        left 90
        forward :width 
        left 90 ]
    end
 
to square :size
    rectangle size size
    end
 
to triangle :size
    repeat 3 [ 
        forward size 
        right 120 ] 
     end
 
to house :size
    left 90
    square size
    triangle size
    right 90
    end
 
to max :lst
   if equalp count lst 1 [ output first lst ]
   make "x max butfirst lst 
   if x > first lst [ output x ] 
   output first lst   
   end
 
to barchart :lst :size
    right 90
    if emptyp lst [ stop ]
    make "scale size / (max lst)
    make "width size / count lst
    foreach lst [ 
        rectangle ? * scale width 
        forward width ]
    back size
    left 90
    end
 
clearscreen hideturtle
house 150
penup
right 90 forward 10 left 90
pendown
barchart [ 0.5 0.33333 2 1.3 0.5 ] 200
left 90 back 10 right 90